 

 

3.4. Язык описания аппартуры Verilog.

3.4.1. Общие сведения

 

Как известно [45], язык описания аппаратуры  Verilog был разработан фирмой  Gateway Design Automation в  1984 .  После поглощения последней таким монстром, как Caddence, язык стал получать все более широкое распространение среди разработчиков и стал не менее популярен, чем VHDL. 

В отличие от VHDL, структура и синтаксис которого напоминают такие «сложные» языки, как АДА или АЛГОЛ, синтаксис Verilog напоминает очень популярный в среде как программистов, так и разработчиков встроенных систем и систем ЦОС старый добрый С. Verilog позволяет провести достаточно эффективно выполнить описание и провести моделирование (simulate) и синтез цифровых схем благодаря применению встроенных (built-in primitives) примитивов, примитивов пользователя (user-defined primitives), средствам временного контроля ( timing checks),  моделирования задержки распространения от входа до выхода (pin-to-pin delay simulation), возможностью задания внешних тестовых сигналов (external stimulus). 

Как и VHDL, Verilog изначально предназначался для моделирования цифровых систем, и как средство описания синтезируемых проектов стал использоваться с 1987 года. В настоящее время ведущие пакеты синтеза систем на ПЛИС, такие как продукты фирм Synopsis, Caddence, Mentor Graphics, многих производителей ПЛИС поддерживают синтез с описания на языке Verilog

В языке Verilog  поддерживается набор типов логических вентилей (Gate Types)

Для логических вентилей определены ключевые слова (Keywords): and (И), nand (И-НЕ), or (ИЛИ) , nor (ИЛИ-НЕ), xor (Исключающее ИЛИ), xnor (Исключающее ИЛИ-НЕ), buf (Буферный элемент), not (Отрицание, НЕ). 

В Verilog при использовании вентилей необходимо задать входы и выходы элемента, а также (не обязательно) имя вентиля . Например, вентили and и  or должны иметь один выход и два и более входов. Так, для вентиля И (and) имеем 

 


and <name><list of arguments>


and myand(out, in1, in2, in3);


and (out, in1, in2

 
Вентили buf и  not могут иметь один вход и один и более выходов. Имя вентиля является необязательным. Примеры использования приведены ниже 

 

        buf mybuf(out1, out2, out3, in);

        not (out, in);

 
Говоря о синтаксисе языка Verilog, следует помнить, что он является контекстно-зависимым языком, то есть строчные и прописные буквы различаются. Все ключевые слова (keywords) задаются в нижнем регистре (строчные буквы). 

Для обозначения пустого пространства (White Space ) используются символы пробела, табуляции и новой строки.
В Verilog поддерживаются два типа комментприев. Они аналогичны принятым в С++ - если надо закомментировать одну строку, то используются две косые черты в ее начапе 

 

 // это комментарий

 

Чтобы закомментировать несколько строк, используются следующая конструкция

 

 /* Это

         комментарии... */

 

Очевидно, что комментарии не могут быть вложенными

 

3.4.2. Операторы 

В Verilog существует три типа операторов – с одним, двумя и тремя операндами. Унарные операторы располагаются слева от операнда, бинарные между операндами и тернарный оператор разделяет три операнда двумя операторами.

Примеры операторов 

 

clock = ~clock; // унарный оператор отрицания



// clock – опреанд

c = a || b; // || - бинарный оператор ИЛИ, a и b операнды

r = s ? t : u;  // ?:  - тернарный оператор, читается как 



//  r = [if s is true then t else u]

 

Как правило, при написании описаний на верилоге комментарии одной строкой (//) используются для ввода текстовых комментариев к коду, а конструкцию /*  */ используют при отладке для «закомментирования» фрагментов кода

 

3.4.3. Числа в Verilog. 

3.4.3.1. Целые числа (Integers)

Целые числа могут быть двоичными (binary), обозначаются  b или B , десятичными (decimal, d или D ), шестнадцатеричными (hexidecimal, h, H ) или восьмеричными (octal, o или O ). Для определения чисел используются следующие форматы:

1. <разрядность>'<основание><число> - полное описание числа

2. <основание><число>  : используется разрядность представления, заданная в системе по определению, но не менее 32 бит. 

3. <число>: используется, когда по умолчанию десятичное основание

Разрядность определяет число бит под представление числа. Например

 

   8'b10100010   // 8 битное число в двоичной системе

   8'hA2         // 8 битное число в шестнадцатиричной системе

 

3.4.3.2. Неопределенное и высокоимпедансное состояния (X and Z values)

Символ xиспользуется для задания неопределенного состояния, символ z показывает третье (высокоимпедансное). При использовании в качестве цифры в числах вместо символа z можно использовать ? . Это рекомендуется делать в операторах выбора (case expressions) для улучшения читаемости кода. Ниже приведены примеры использования символов x и z в числах.

 

4'b10x0

4'b101z

12'dz

12'd? 

8'h4x     

 

3.4.3.4. Отрицательные числа (Negative numbers )

Отрицательное число задается с помощью знака минус перед разрядностью числа. Примеры отрицательных чисел

 

-8'd5  

 8'b-5  // неправильно, знак минус перед числом, а не разорядностью!

 

3.4.3.5. Символ подчеркивания (Underscore )

Знак подчеркивания (_, underscores) может быть записан в любом месте числа,что позволяет использовать его как разделитель разрядов, улучшающий читабельность. 

 

16'b0001_1010_1000_1111  // использование подчеркивания

8'b_0001_1010     // некорректное использование подчеркивания

 

3.4.3.6. Действительные числа (Real )

Действительные числа (Real numbers) могут быть представлены либо в десятичном виде, либо в стандартной форме с плавающей точкой ( scientific format). Примеры действительных чисел.

        1.8

        3_2387.3398_3047

        3.8e10      // e или E для обозначения порядка

        2.1e-9

        3.   // неправильно!

3.4.3.7. Строки (Strings )

Строка заключается в кавычки " ... " и не могут занимать более  одной линии. Примеры использования строк.

 

        "hello world";   // правильное использование строк

        "good
           b
            y
             e
           wo

            rld";        // неправильное использование строк
 

3.4.4.Цепи в Verilog (Nets)

 
Для обозначения цепей используются следующие ключевые слова: wire, supply0, supply1. Величина по умолчанию (default value): z. Разрядность по умолчанию (default size): 1 бит 

По своему назначению цепь (Net) в Verilog сходна с сигналом в VHDL.. Цепи обеспечивают непрерывное модифицирование сигналов на выходах цифровой схемы относительно изменения сигналов на ее входах. 

Если драйвер (источник) сигнала цепи  имеет некоторое значение, то и цепь принимает то же значение. Если драйверы (drivers) цепи принимают различные значения, цепь принимает значение наиболее «сильного»  сигнала (strongest), если же «сила» каждого сигнала равнозначна, то цепь принимает неопределенное состояние (x). 

Для обозначения цепи наиболее часто применяется ключевое слово wire, ключевые слова supply0 и supply1, используются для моделирования источников питания (power supplies) в схеме.

 

3.4.5. Регистры (Registers )

Для обозначения регистров используются следующие ключевые слова reg. Величина по умолчанию: х. Разрядность по умолчанию: 1 бит 

 

Основное различие между цепями (nets) и регистрами (registers) состоит в том, что значение регистра должно быть назначено явно. Эта величина сохраняется до тех пор, пока не сделано новое назначение. Рассмотрим использование этого свойство на примере триггера с разрешением (защелки, E-type flip flop)

 

module E_ff(q, data, enable, reset, clock);

            output q;

            input data, enable, reset, clock;

            reg q;

 

        always @(posedge clock)

                if (reset == 0)

                 q = 1'b0;

                else if (enable==1)

                 q = data;

        endmodule

 

Регистр q хранит записанную в него величину до тех пор, пока не произойдет нового назначения сигнала 

Для того, чтобы провести моделирование этого примера, напишем модуль верхнего уровня (higher level module), который формирует тестовый сигнал и позволяет провести наблюдение за выходами триггера. В этом случае сигнал q является цепью, драйвером (источником) которой служит модуль E_ff .

 

        module stimulus;

          reg data, enable, clock, reset;

          wire q;

 

        initial begin

          clock = 1'b0;

          forever #5 clock = ~clock;

        end

 

        E_ff eff0(q, data, enable, reset, clock);

 

        initial begin

          reset = 1'b0;

          #10 reset = 1'b1;

              data = 1'b1;

          #20 enable = 1;

          #10 data = 1'b0;

          #10 data = 1'b1;

          #10 enable = 0;

          #10 data = 1'b0;

          #10 data = 1'b1;

          #10 enable = 1;

          #10 reset = 1'b0;

          #30 $finish;

        end

 

        initial

          $monitor($time, " q = %d", q);

 

        endmodule
 
 
3.4.6. Векторы (Vectors )

Как цепи, так и регистры могут иметь произвольную разрядность, если их объявлять как векторы. Ниже приведены примеры объявлений векторов

 

        reg [3:0] output; // выход 4-разрядного регистра

        wire [31:0] data; // 32 битная цепь

        reg [7:0] a;

 

        data[3:0] = output; // частичное назначение

        output = 4'b0101;   // Назначение на целый регистр

 

Важным является порядок назначения элементов в векторе. Первый элемент регистра является наиболее значимым (most significant)

 

        reg [3:0] a;  // а3 – старший бит

        reg [0:3] b;  // b0- старший бит.

 

3.4.7. Массивы (Arrays )

 

Регистры (Registers), целые числа (integers) и временные (time) типы данных можно объявлять как массивы, как это показано в нижеследующем примере: 

Объявление
 <data_type_spec> {size} <variable_name> {array_size} 

 

Использование
 <variable_name> {array_reference} {bit_reference} 

 

        reg data [7:0]; // 8 1-разрядных элементов

 

        integer [3:0] out [31:0]; // 32 4-битных элемента

 

        data[5]; // 5 8 битных элементов

 

3.4.8. Регистровые файлы (Memories )

 

Регистровый файл (Memories) представляет собой массив регистров (array of registers). Ниже представлен синтаксис объявления регистрового файла.. 

 

reg [15:0] mem16_1024 [1023:0]; // регистровый файл 1K Х 16

 

mem16_1024[489]; // 489 элемент файла mem16_1024
 

Желательно для обозначения регистровых файлов использовать информативные имена, например mem16_1024 чтобы избежать путаницы.

 

3.4.9. Элементы с третьим состоянием (Tri-state)

 

Как уже упоминалось, в языке Verilog ситуация, когда драйверами одной цепи являются более одного источника с разными значениями, разрешается в пользу источника, имеющего большую «силу» ( signal "strengths").  Наименьшую «силу»  имеет сигнал z, обозночающий третье или высокоимпедансное (high-impedance) состояние.  Правда эти рассуждения актуалны именно на этапе моделирования, но никак не синтеза ПЛИС, о них следует помнить при разработке тестов. Таким образом драйвер в третьем состоянии не влияет на итоговое значение сигнала цепи. Пример драйвера с третьим состоянием приведен ниже

 

        module triDriver(bus, drive, value);

           inout [3:0] bus;

           input       drive;

           input [3:0] value;

 

           assign #2 bus = (drive == 1) ? value : 4'bz;

 

        endmodule // triDriver

 

В данном примере, когда управляющий сигнал принимает высокий уровень, шина принимает значение входной величины, в протиивном случае шина переходит в третье состояние. 

В следующем примере используются несколько управляющих комбинаций для трех буферов с тремя состояниями

 

module myTest;

     wire [3:0] bus;

     reg          drive0, drive1, drive2;

 

     integer    i;

 

     triDriver mod1 (bus, drive0, i[3:0]);

     triDriver mod2 (bus, drive1, 4'hf);

     triDriver mod3 (bus, drive2, 4'h0);

 

     initial begin

          for (i = 0; i < 12; i = i + 1) begin

           #5 {drive2, drive1, drive0} = i;

           #5 $display ($time,"  %b %b %d", i[2:0], bus, bus);

          end

        $finish;

     end 

 

   endmodule 

 

Результаты прогона модели имеют вид

 

                   10  000 zzzz  z

                   20  001 0001  1

                   30  010 1111 15

                   40  011 xx11  X

                   50  100 0000  0

                   60  101 0x0x  X

                   70  110 xxxx  x

                   80  111 xxxx  x

                   90  000 zzzz  z

                  100  001 1001  9

                  110  010 1111 15

                  120  011 1x11  X

 

3.4.10. Арифметические операторы (Arithmetic operators )

 

Символы (keysymbols)  *, /, +, -, % 

 

Бинарные операторы умножения, деления, сложения, вычитания, определения остатка от деления представлены в нижеследующем примере. 

 

        module arithTest;

           reg [3:0] a, b;

 

           initial begin

              a = 4'b1100; // 12

              b = 4'b0011; // 3

 

              $displayb(a * b); // умножение - 4'b1000

                                // 4 МСБ
              $display(a / b);  // деление 4

              $display(a + b);  // сложение 15

              $display(a - b);  // вычитание 9

              $display((a + 1'b1) % b); // остаток 1

 

           end

 

        endmodule // arithTest

 

Унарные плюс и минус имеют более высокий приоритет (precedance), чем бинарные операторы. 

Следует заметить, что если хотя бы один бит в одном из операндов неопределен (равен x), то и результат операции также будет неопределен. 

 

3.4.11. Логические операторы (Logical Operators)

 

Ключевые символы: &&, ||, !. 

К логическим операциям относятся И (and), ИЛИ ( or) и НЕ (not). Результат логической операции может принимать значения истинно (true,  1 ) или ложно ( false, 0 ), а также иметь неопределенное состояние (unknown, x).  При выполнении логического оператора все неопределенные величины, как и операнды в третьем состоянии принимаются как имеющие низкий логический уровень (false). В качестве операнда может выступать как переменная (variable), так и логическое выражение (expression).  Пример работы логических операторов приведен ниже

 

        module logicalTest;

 

           reg [3:0] a, b, c;

 

           initial begin

              a = 2; b = 0; c = 4'hx;

 

              $display(a && b); // И 0

              $display(a || b); // ИЛИ 1

              $display(!a); // НЕ 0

              $display(a || c); // 1, unknown || 1 (=1)

              $display(!c); // unknown

              end

 

        endmodule // logicalTest

 

3.4.12. Операторы отношения (Relational Operators)

 

Ключевые символы: >, <, >=, <=. 

К оператора м отношения относятся операторы «больше», «меньше», «больше или равно», «меньше или равно». Результат операции – истина или ложь. Если хотя бы один операнд неопределен, то и результат операции будет неопределен.

 

        module relatTest;

           reg [3:0] a, b ,c, d;

 

           initial begin

              a=2;

              b=5;

              c=2;

              d=4'hx;

 

              $display(a < b);  // true, 1

              $display(a > b); false, 0

              $display(a >= c); // true, 1

              $display(d <= a); unknown

            end

        endmodule // relatTest
 

3.4.13. Операторы эквивалентности (Equality )

 

Ключевые символы: ==, !=, ===, !==. 

К операторам эквивалентности (equality operators) относятся оператор логического равенства (logical equality), неравенства ( logical inequality), выборочного равенства (case equality) и неравенства (inequality).  Эти операторы сравнивают операнды побитно. Логические операторы возвращают неопределенный результат, если операнд содержит неопределенные биты, в отличие от выборочных операторов. В случае неравной длины операндов, более короткий дополняется нулями.

Ниже приведен пример использования операторов эквивалентности. 

 

module equTest;

           reg [3:0] a, b ,c, d, e, f;

 

           initial begin

              a = 4; b = 7;    // these default to decimal bases

              c = 4'b010;

              d = 4'bx10;

              e = 4'bx101;

              f = 4'bxx01;

 

              $displayb(c);      // outputs 0010

              $displayb(d);      // outputs xx10

 

              $display(a == b);  // logical equality, evaluates to 0

              $display(c != d);  // logical inequality, evaluates to x

              $display(c != f);  // logical inequality, evaluates to 1

              $display(d === e); // case equality, evaluates to 0

              $display(c !== d); // case inequality, evaluates to 1

           end

        endmodule // equTest

 

Ну и наверное совершенно понятно, что операторы эквивалентности и присваивания – совершенно различные операторы.
 

3.4.14. Поразрядные Операторы (Bitwise Operators )

 

Ключевые символы: ~, &, |, ^, (~^, ^~). 

 

К поразрядным операторам относятся поразрядное отрицание, поразрядные логические И, ИЛИ, исключающее или, исключающее ИЛИ-НЕ. Поразрядные операторы выполняются только над операндами, имеющими одинаковую разрядность. В том случае, если разрядность одного операнда меньше другого, недостающие разряды дополняются нулямиы . Ниже приведен пример использования поразрядных операторов. 

 

        module bitTest;

           reg [3:0] a, b ,c;

 

  initial begin

  a = 4'b1100; b = 4'b0011; c = 4'b0101;

  $displayb(~a);     // bitwise negation, evaluates to 4'b0011

  $displayb(a & c);  // bitwise and, evaluates to 4'b0100

  $displayb(a | b);  // bitwise or, evaluates to 4'b1111

  $displayb(b ^ c);  // bitwise xor, evaluates to 4'b0110

  $displayb(a ~^ c); // bitwise xnor, evaluates to 4'b0110

  end

  endmodule // bitTest

 

3.4.15.Операторы приведения (Reduction Operator )

Keysymbols: &, ~&, |, ~|, ^, ~^, ^~. 

Операторы приведения – И, ИЛИ, И-НЕ, ИЛИ-НЕ, исключающее или, исключающее ИЛИ-НЕ  (два варианта). Они выполняются над многоразрядным операндом пошагово, бит за битом, начиная с двух крайних левых разрядов, выдавая на выходе одноразрядный результат. Очевидно, что такой подход позволяет реализовать проверку на четность (нечетность). Ниже приведены примеры использования операторов приведения.

 

        module reductTest;

           reg [3:0] a, b ,c;

 

           initial begin

              a = 4'b1111;

              b = 4'b0101;

              c = 4'b0011;

 

$displayb(& a);//, (то же 1&1&1&1), равен 1 1

$displayb(| b); // (same as 0|1|0|1), evaluates to 1 $displayb(^ b); // искл.ИЛИ (same as 0^1^0^1), evaluates to 0

           end

 

        endmodule // reductTest

Безусловно, следует улавливать различия между логическими операторами, поразрядными операторами и операторами приведения. Несмотря на схожесть символов этих операторов, число операндов в каждом случае различно. 

 

3.4.16.Операторы сдвига (Shift Operator). 

Ключевые символы: >>, <<. 

Операторы сдвига позволяют осуществить сдвиг операнда как вправо, так и влево. Пример их использования приведен ниже 

 

        module shiftTest;

           reg [3:0] a;

 

           initial begin

              a = 4'b1010;

 

$displayb(a << 1);   // shift left by 1, evaluates to 4'b0100

$displayb(a >> 2);   // shift right by 2, evaluates to 4'b0010

           end

 

        endmodule // shiftTest

 

Этот оператор часто применяют для реализации регистров сдвига, длинных алгоритмов перемножения и т.п.

3.4.17. Конкатенация (объединение,Concatenation )

Ключевой символ: {, } 

Объединение позволяет увеличить разрядность (size) цепей (nets), регистров (registers) и т.д.

        module concatTest;

           reg         a;

           reg [1:0] b;

           reg [5:0] c;

 

           initial begin

              a = 1'b1;

              b = 2'b00;

              c = 6'b101001;

 

$displayb({a, b});  // produces a 3-bit number 3'b100

$displayb({c[5:3], a}); // produces 4-bit number 4'b1011

 

           end

 

        endmodule // concatTest

 

3.4.18. Повторение (Replication )

Повторение (Replication) может быть использовано для многократного повторения объединения (concatenation), как показано в нижеследующем примере. 

 

        module replicTest;

           reg         a;

           reg [1:0] b;

           reg [5:0] c;

 

           initial begin

              a = 1'b1;

              b = 2'b00;

 

              $displayb({4{a}}); // результат -  1111

              c = {4{a}};

              $displayb(c);     // результат -001111

           end

 

        endmodule // replicTest

 

 

 

3.4.19. Системные директивы (System Tasks )

 

Наверное, этот раздел не будет интересен тем, кто пишет только синтезируемые описания на Verilog. Но поскольку язык является не только средством описания проекта, но и довольно мощным инструментом для поведенческого моделирования систем, то следует сказать несколько слов о встроенных директивах компилятора, позволяющих выполнить моделирование и произвести анализ его результатов.

 

Директивы вывода результатов моделирования (Writing to Standard Output)

 

Ключевые слова: $display, $displayb, $displayh, $displayo, $write, $writeb, $writeh, $writeo. 

 

Наиболее часто применяется директива $display. Она может быть использована для вывода на экран строк, выражений или переменных. Ниже приведен пример использования директивы $display 

 

        $display("Hello Dr Blair");

        --- output: Hello Dr Blair

 

        $display($time) // current simulation time.

        --- output: 460

 

        counter = 4'b10;

        $display(" The count is %b", counter);

        --- output:  The count is 0010

 

Синтакс определения формата вывода аналогичен синтаксису printf в языке программирования C. Ниже приведено его описание для директивы  $display 

 

	Формат
	Описание

	%d or %D
%b or %B

%h or %H

%o or %O

%m or %M

%t or %T

%e or %E

%f or %F

%g or %G
	Decimal

Binary

Hexadecimal

Octal

Hierarchical name

Time format

Real in scientific format

Real in decimal formal

Real in shorter of above two


Для специальных символов используются следующие эскейп-последовательности (escape sequence)

 

	\n
\t
\\

\"

%%
	Новая строка

табуляция

\

"

%


 

Директива $write идентична директиве $display, за исключением того, что она не осуществляет автоматический переход на новую строку в конце вывода информации

Если спецификация вывода не определена, то по умолчанию исполшьзуются следующие форматы 

 

	Директива
	Формат по умолчанию

	$display

$displayb

$displayh

$displayo

$write

$writeb

$writeh

$writeo
	decimal

binary

hexadecimal

octal

decimal

binary

hexadecimal

octal


 

Так, например,  следующий фрагмент кода …

              $write(5'b01101);

              $writeb("  ", 5'b01101);

              $writeh("  ", 5'b01101);

              $writeo("  ", 5'b01101,"\n");

 

… И  результат его работы 

13 13    01101  0d  15

14 14     

 

3.4.20. Контроль процесса моделирования (Monitoring a Simulation). 

 

Ключевые слова: $monitor, $monitoron, $monitoroff 

Формат директивы  $monitor практически аналогичен формату $display. Разница заключается в том, что выход формируется при любом изменении переменных, которое произойдет в оперделенное время . Наблюдение может быть включено или отключено с помощью директив $monitoron или $monitoroff соответственно. По умолчанию в начале моделирования наблюдение за его ходом включено. Ниже приведен пример наюлюдения за моделированием. 

 

        module myTest;

           integer a,b;

 

           initial begin

              a = 2;

              b = 4;

              forever begin

              #5 a = a + b;

              #5 b = a - 1;

              end // forever begin

           end // initial begin

 

           initial #40 $finish;

 

           initial begin

              $monitor($time, " a = %d, b = %d", a, b);

           end // initial begin

 

        endmodule // myTest


Результат прогона  

                           0 a =           2, b =           4

                           5 a =           6, b =           4

                          10 a =           6, b =           5

                          15 a =          11, b =           5

                          20 a =          11, b =          10

                          25 a =          21, b =          10

                          30 a =          21, b =          20

                          35 a =          41, b =          20

 

3.4.21. Окончание моделирования (Ending a simulation)

 

Ключевые слова: $stop, $finish. 

Директива $finish завершает симуляцию и передает управление операционной системе. Директива. $stop приостанавливает моделирование и переводит систему с Verilog в интерактивный режим (см. пример ниже).

 

 

        initial begin

          clock = 1'b0;

          #200 $stop   

          #500 $finish         End

 

3.4.22. Проектирование комбинационных схем: пример проектирования мультиплексора  4 в 1.

 

Рассмотрим проектирование комбинационных логических устройств на примере мультиплексора 4 в 1 

При проектировании мультиплексора рассмотрим различные методы его построения 

 

3.4.22.1. Реализация на уровне логических вентилей (Gate Level Implementation)

 

Рассмотрим пример реализации нашего мультиплексора 4 в 1 на  уровне логических вентилей. 

 

module multiplexor4_1(out, in1, in2, in3, in4, cntrl1, cntrl2);

 

          output out;

          input in1, in2, in3, in4, cntrl1, cntrl2;

          wire notcntlr1, notcntrl2, w, x, y, z;

 

          not (notcntrl1, cntrl1);

          not (notcntrl2, cntrl2);

 

          and (w, in1, notcntrl1, notcntrl2);

          and (x, in2, notcntrl1, cntrl2);

          and (y, in3, cntrl1, notcntrl2);

          and (z, in4, cntrl1, cntrl2);

 

          or (out, w, x, y, z);

 

endmodule

 

Начнем подробное построчное рассмотрение этого примера. Итак, «во первых строках» у нас … 

 

module multiplexor4_1(out, in1, in2, in3, in4, cntrl1, cntrl2);

 

Первая строка – описание модуля, ключевое слово – module – используется вместе с именем модуля, по которму осуществляется ссылка на модуль. В скобках приведен список портов модуля (port list) причем вначале перечисляются выходы, затем входы. Каждая строка завершается точкой с запятой – это, как известно святое для многих языков высокого уровня.

 

        output out;

        input in1, in2, in3, in4, cntrl1, cntrl2;
 

Все порты в списке должны быть объявлены как входы (input), выходы (output) или двунаправленые выводы (inout), в этом случае они по умолчанию назначаются типом цепь (wire), если нет других указаний. Когда назнаено имя цепи система моделирования на базе Verilog ожидает неявное назначение выходного сигнала, оценивая его чтобы осуществлять передачу этого сигнала к внешним модулям.

 

         wire notcntrl1, notcntrl2, w, x, y, z;

 

В данной строке опредекляются внутренние цепи, осуществляющие объединение узлов мультиплексора. Как видим понятия цепи в Verilog и сигнала в VHDL очень схожи

 

        not (notcntrl1, cntrl1);

        not (notcntrl2, cntrl2);


В этих строчках описывается вентили НЕ с входами cntrl1 и cntrl2 и выходами notcntrl1 и notcntrl2 соответственно. Следует помнить, что в описании портов вентиля всегда вначале идут выходы, затем входы

 

          and (w, in1, notcntrl1, notcntrl2);

          and (x, in2, notcntrl1, cntrl2);

          and (y, in3, cntrl1, notcntrl2);

          and (z, in4, cntrl1, cntrl2);

 

          or (out, w, x, y, z);

Аналогично описываются и вентили И и ИЛИ

endmodule 


Конец модуля завершается ключевым словом  endmodule. 

 

3.4.22.2.Реализация мультиплексора с помощью логических операторов (Logic Statement Implementation)

 

Реализация мультиплексора с использованием логических операторов имеет вид

        module multiplexor4_1 (out, in1, in2, in3 ,in4, cntrl1, cntrl2);

           output out;

           input in1, in2, in3, in4, cntrl1, cntrl2;

 

           assign out = (in1 & ~cntrl1 & ~cntrl2) |

                        (in2 & ~cntrl1 &  cntrl2) |

                        (in3 &  cntrl1 & ~cntrl2) |

                        (in4 &  cntrl1 &  cntrl2);

        endmodule


Ну в начале все понятно – имя проекта, порты … 

        module multiplexor4_1 (out, in1, in2, in3 ,in4, cntrl1, cntrl2);

          output out;

          input in1, in2, in3, in4, cntrl1, cntrl2;


Обратим внимание, что ни порты, ни их количество не изменились – стыковка с внешними модулями также останется неизменной. В какой то мере это напоминает проект с использованием VHDL, когда для одного интерфейса можно описать несколько архитектурных тел.. 

 

        assign out = (in1 & ~cntrl1 & ~cntrl2) |

                     (in2 & ~cntrl1 &  cntrl2) |

                     (in3 &  cntrl1 & ~cntrl2) |

                     (in4 &  cntrl1 &  cntrl2);

        endmodule


С помощью конструкции assign  осуществляется непрерывное назначение (continuous assignment) цепи out. В этом случае ее значение заново вычисляется каждый раз, когда меняется хоть один из операндов.

 

3.4.22.3. Реализация с помощью оператора выбора (Case Statement Implementation )

 

Рассмотрим использование оператора выбора (case statement) для реализации мультиплексора. Следует заметить, что этот способ наверное наиболее прост и наименее трудоемок..

 

        module multiplexor4_1 (out, in1, in2, in3, in4, cntrl1, cntrl2);

           output out;

           input  in1, in2, in3, in4, cntrl1, cntrl2;

           reg      out;

 

           always @(in1 or in2 or in3 or in4 or cntrl1 or cntrl2)

              case ({cntrl1, cntrl2})

             2'b00 : out = in1;

             2'b01 : out = in2;

             2'b10 : out = in3;

             2'b11 : out = in4;

             default : $display("Please check control bits");

              endcase

        endmodule


Ну первые три строки до боли знакомы – добавить к сказанному ранее практически нечего. 

        module multiplexor4_1 (out, in1, in2, in3, in4, cntrl1, cntrl2);

          output out;

          input in1, in2, in3, in4, cntrl1, cntrl2;

          reg out;


Единственное отличие – выход out определен как регистр (register), это сделано для того, чтобы назначать его значения явно и не управлять ими,такое назначение сигнала называется  процедурным назначением (procedural assignment).  Данные типа цепь (wire) не могут быть назначены явно, они нуждаются в сигнале драйвере, такое назначение называется непрерывным назначением (continuous assignment) 

 

        always @(in1 or in2 or in3 or in4 or cntrl1 or cntrl2)


Эта конструкция читается так же как и пишется, то есть значение вычисляется всегда при изменении хотя бы одного операнда – система постоянно их отслеживает. Несколько забегая вперед, следует отметить, что данная конструкция является синтезируемой во многих системах проектирования, в частности и в  Max + Plus  фирмы  Altera. Список переменных называется списком чувствительности (sensitivity list), поскольку данная конструкция чувст вительна к их изменениям. Данный термин нам уже известен из описания языка VHDL
 

Ключевое слово always @( expr or expr ... ); 

 

              case ({cntrl2, cntrl1})

             2'b00 : out = in1;

             2'b01 : out = in2;

             2'b10 : out = in3;

             2'b11 : out = in4;

             default : $display("Please check control bits");

              endcase

        endmodule

 

Синтаксис  оператора выбора case в Verilog  сходен с синтаксисом оператора выбора в языке  C. Условием является контатенация или объединение переменных cntr2 и cntr1 в двухразрядное число. Завершает оператор выбора лючевое слово  endcase. 

Здесь нелишне напомнить, что следует различать процедурное и непрерывное назначение сигналов, что и иллюстрируют вышеприведенные примеры 

 

3.4.22.4. Реализация с использованием условного оператора (Conditional Operator Implementation )

 

В принципе ничего нового мы уже не видим – ясно видно, что в нашем случае условный оператор проигрывает оператору case  в наглядности представления (хотя для большинства систем проектирования синтезируемые реализации окажуться идентичными)

 

        module  multiplexor4_1 (out, in1, in2, in3, in4, cntrl1, cntrl2);

          output out;

          input in1, in2, in3, in4, cntrl1, cntrl2;

 

          assign out = cntrl1 ? (cntrl2 ? in4 : in3) : (cntrl2 ? in2 : in1);

 

        endmodule

 

3.4.22.5. Тестовый модуль (The Stimulus Module )

Вообще говоря, первоначально язык Verilog задумывался как язык верификации цифровых устройств (Verify Logic)Поэтому одной из мощных возможностей языка является наличие средств для задания тестовых сигналов. Ниже приводиться пример такого модуля для нашего мультиплексора 4 в 1.

 

 

 

        module muxstimulus;

          reg IN1, IN2, IN3, IN4, CNTRL1, CNTRL2;

          wire OUT;

 

        multiplexor4_1 mux1_4(OUT, IN1, IN2, IN3, IN4, CNTRL1, CNTRL2);

 

        initial begin

          IN1 = 1; IN2 = 0; IN3 = 1; IN4 = 0;

          $display("Initial arbitrary values");

          #0 $display("input1 = %b, input2 = %b, input3 = %b, input4 = %b\n",

                                IN1, IN2, IN3, IN4);

 

         {CNTRL1, CNTRL2} = 2'b00;

         #1 $display("cntrl1=%b, cntrl2=%b, output is %b", CNTRL1, CNTRL2, OUT);

 

           {CNTRL1, CNTRL2} = 2'b01;

         #1 $display("cntrl1=%b, cntrl2=%b output is %b", CNTRL1, CNTRL2, OUT);

 

    {CNTRL1, CNTRL2} = 2'b10;

#1 $display("cntrl1=%b, cntrl2=%b output is %b", CNTRL1, CNTRL2, OUT);

 

    {CNTRL1, CNTRL2} = 2'b11;

#1 $display("cntrl1=%b, cntrl2=%b output is %b", CNTRL1, CNTRL2, OUT);

 

        end

 

        endmodule

 

Рассмотрим данный пример подробнее. 

 

        module muxstimulus;

 

Поскольку наш генератор тестов является модулем верхнего уровня иерархии (top level module), то в его описании отсутствует список портов. Для задния модуля используется ключевое словп module. 

 

          reg IN1, IN2, IN3, IN4, CNTRL1, CNTRL2;

          wire OUT;
 

В данном случае мы обеспечиваем подачу тестовых сигналов на входы нашего мультиплексора и  снятие сигнала с его выхода, следовательно, мы должны назначить сигналы для входов мультиплексора и сигналу, который управляется (driven) его выходом. Поэтому и использованы данные типа reg для входов и  wire для выхода. 

 

multiplexor4_1 mux1_4(OUT, IN1, IN2, IN3, IN4, CNTRL1, CNTRL2);

 

В  этой строке происходит обращение к тестируемому модулю  multiplexor4_1 , в таких случаях принят следующий синтакс

 

<module_name> <instance_name> (port list); 

 

Имя экземпляра (instance name) очень похоже на понятие языка VHDL и является необходимым при обращении к определяемым пользователем модулям (user defined modules) дабы не нарушать иерархие проекта

Список портов (port list) устанавливает соответствие между сигналами тестового и тестируемого модулей, при этом порядок переменных в списке портов должен соответствовать порядку их объявления в тестовом модуле.

        initial begin

          IN1 = 1; IN2 = 0; IN3 = 1; IN4 = 0;

          $display("Initial arbitrary values");

          #0 $display("input1 = %b, input2 = %b, input3 = %b, input4 = %b\n",

                                IN1, IN2, IN3, IN4);

 

Собственно моделирование (simulation) осуществляется конструкцией, определяемой ключевыми словами initial begin ... end. Она предназначена для объединения инструкций, которые могут выполняться одновременно. 

Перед инициализацией процесса моделирования сообщение об этом выводиться с помощью $display. Конструкция  #0 перед директивой вывода означает, что вывод на экран осуществляется после назначения сигналов на входы.  Синтакс директивы $display подобен синтаксису функции printf в языке C, 

 

$display( expr1, expr2, ...., exprN); 

Expr N может быть переменной, выражением или строкой. 

 

        {CNTRL1, CNTRL2} = 2'b00;

         #1 $display("cntrl1=%b, cntrl2=%b, output is %b", CNTRL1, CNTRL2, OUT);

 

Здесь осуществляется назначение сигналов управления  

 

CNTRL1 = 0; 
CNTRL2 = 0; 



Оператор  конкатенации (concatenation operator) { } может быть использован для группового назначения. Следует помнить, что разрядов число в числе и переменной должно совпадать

 

{CNTRL1, CNTRL2} = 2'b01;

#1 $display("cntrl1=%b, cntrl2=%b output is %b", CNTRL1, CNTRL2, OUT);

 

{CNTRL1, CNTRL2} = 2'b10;

#1 $display("cntrl1=%b, cntrl2=%b output is %b", CNTRL1, CNTRL2, OUT);

 

{CNTRL1, CNTRL2} = 2'b11;

#1 $display("cntrl1=%b, cntrl2=%b output is %b", CNTRL1, CNTRL2, OUT);

 

end

endmodule
 

В этом фрагменте сигналы управления изменяются и изменение выхода отслеживается директивой display.

В этом примере дипектива  $display выполняется с некоторой задержкой относительно сигналов управления, что позволяет отобразить корректные значения. 

 

3.4.24. Модули проекта (Design Blocks Modules)

Расммотрим правила построения иерархического проекта. В качестве примера мы будем рассматривать триггер с разрешением (E-type flip flop). В Verilog возможны два подхода для его реализации

Сверху вниз (Topdown)- начать с описания Е-триггера и далее добавлять детали преокта

Снизу вверх (Bottom up) – начать с базовых блоков, а затем объединить их в единый проект

 

Мы воспользуемся технологией снизу вверх, хртя наверное, наиболее удобно использование обоих подходов.

 

Начнем с описания D- триггера как модуля. Ключевые слова для задания модуля 

 

module <module_name> ... endmodule, as below. 

 

Ниже приведено его описание

 

        module dff(q, data, clock);

           output q;

           input  data, clock;

           reg      q;

 

           always @(posedge clock)

              q = data;

        endmodule // dff

 

Аналогично поведенческое описание мультплексора с инверсией (inverting multiplexor) имеет вид 

 

module mux2_1(out, in1, in2, cntrl);

           output out;

           input  in1, in2, cntrl;

 

           assign out = cntrl ? ~in1 : ~in2;

        endmodule // mux2_1

 

Тогда описание модуля верхнего уровня для Е-триггера, состоящего из D- триггера и мультиплексора принмиает вид

 

        module e_ff(q, data, enable, reset, clock);

           output q;

           input  data, enable, reset, clock;

           wire     norout, muxout;

 

           mux2_1 mod1 (muxout, data, q, enable);

           nor         (norout, reset, muxout);

           dff    dff0 (q, norout, clock);

        endmodule

Обратим еще раз внимаение на обращение к модулю, определяемому пользователем – через имя экземпляра модуля

 

name_of_module instance_name (port_list); 

 

 

Ниже приведен текст модуля для тестирования нашего триггера. Как и в примере с мультиплексором, он не имеет списка портов, поскольку является самым верхним в иерархии

 

        module e_ffStimulus;

          reg data, enable, reset, clock;

          wire q;

 

          e_ff mod1 (q, data, enable, reset, clock);

 

          initial begin

             clock = 1'b0;

             forever  clock = #5 ~clock;

          end

 

          initial begin

             enable = 1'b0;     // переменная разрешения
             reset = 1'b1;      // активный уровень сброса - 1

             #20 reset = 1'b0;  // сьрос.

                 data = 1'b1;   установка в  HIGH

             #10 enable = 1'b1; // and then enable data latching

             #10 data = 1'b1;   // изменение данных

             #20 data = 1'b0;   // изменение данных

             #30 data = 1'b1;   // изменение данных

             #10 data = 1'b0;   // изменение данных

             #10 data = 1'b1;   // изменение данных

             #20 enable = 1'b0; // запрет защелкивания

             #10 data = 1'b0;   // изменение данных 

             #10 reset = 1'b1;  // сьрос снова

             #20 $finish;      // все

          end                  // такт остановлен 

          initial begin

             $display($time, " reset, enable, data,  q ");

             $display($time, "   %d      %d     %d     %d",

                    reset, enable, data, q);

             forever #10 $display($time, "   %d      %d     %d     %d",

                    reset, enable, data, q);

          end

 

        endmodule // e_ffStimulus

 

3.4.25. Порты (Ports)

 

Порты обеспечивают связь модуля с другими модулями проекта.Ниже приведен порты для нашего примера

 

module d_ff(q, d, reset, clock);

module e_ff(q, d, enable, reset, clock); 

module Stimulus; 

 

Каждый порт может  быть объявлен как вход (input), выход (output) или двунаправлекнный (inout). Все объявленные порты по умолчанию назначаются цепи (wire), для изменения типа сигнала необходимо выполнить назначение отдельно. Например
 

        module d_ff(q, d, reset, clock);

             output q;      // обязательное объявление портов 

             input d, reset, clock; // как входы и выходы

             reg q;  // снова объявляем как регистр

 

Не забываем, что выходы модуля идут первыми в списке портов. 

 

3.4.26. Правила соединения (Connection Rules )

 

Выше мы рассматривали два типа модулей  - внешние и внутренние (outer and inner ), в нашем примере внешним модулем был Е-триггер, внутренним – триггер типа D. Ниже рассмотрим основные правила соелдинения модулей

 Входы (Inputs )

Входы внутреннего модуля всегда должны имет тип цепь, поскольку они управляются внешними сигналами. Входы внешнего модуля могуьт иметь тип как цепь, так и регистр

 

 Выходы (Outputs )

Во внутреннем модуле выходы могут иметь тип как цепь, так и регистр, в то время как выходы внешнего модуля дожны быть типа цепь, поскольку управляются внутренними модулями

 

 Двунаправленые выводы (Inouts )

Двунаправленный порт всегда имеет тип цепь.

 

 Соответствие портов (Port Matching)

Очевидно, обращаясь к модулю, необходимо, чтобы совпадала разрядность портов.

 

 Присоединение портов (Connecting Ports )

Присоединение портов осуществляется либо по порядку (ordered list) или по имени порта (or by name). Например
 

        module d_ff( q, d, reset, clock);

             ...

        endmodule

 

        module e_ff(q, d , enable, reset, clock);

             output q;

             input d, enable, reset, clock;

             wire inD;

                    ...

             d_ff dff0(q, inD, reset, clock);

                    ...

        endmodule
 

 

3.4.27. Базовые блоки (Basic Blocks )

В Verilog  приняты два типа назначений – непрерывное и процедурное. Непрерывное назначение может быть выполнено для цепей nets или для их объединений (concatenation of nets). Операнды могут иметь произвольный тип данных. 

Процедурное назначение может быть выполнено для данных типов reg, integer, real или time.

 

3.4.28. Инициализация (Initial Block )

Ключевые слова: initial 

Блок инициализации состоит из группы операторов, заключенных в операторные скобки  begin... end и которые будут выполняться с момента старта моделирования. Ниже приведен пример инициализации моделирования

 

        initial

        clock = 1'b0;  

 

        initial

           begin  

           alpha = 0;

           #10 alpha = 1;  // генерация сигнала

           #20 alpha = 0;

           #5  alpha = 1;

           #7  alpha = 0;

           #10 alpha = 1;

           #20 alpha = 0;

           end;

 

3.4.29. Конструкция always (Always Block )

 

Ключевые слова: always 

Блок  always означает, что действие ыполняется  непрерывно, пока процесс моделирования не будет остановлен директивойц $finish or $stop. 

Note: the $finish command actually terminates the simulation where as $stop. merely pauses Ниже приведен пример формирования тактового импульса с использованием конструкции always . Заметим, что вообще говоря для этих целей удобнее использовать циклы

 

        module pulse;

 

        reg clock;

 

        initial clock = 1'b0; // запуск такта
        always #10 clock = ~clock; // такт
        initial #5000 $finish // конец моделирования

 

        endmodule

 

3.4.30. Пример проектирования последовательностного устройства: двоичный счетчик 

 

Рассмотрим пример проектирования двоичного счетчика считающего от 0 до 12 и по достижении 12 сбрасывающегося в 0 и начинающего счет заново. 

Рассмотрим описание вентильном уровне

 

        // 4-bit binary counter

 

        module counter4_bit(q, d, increment, load_data, global_reset, clock);

           output [3:0]     q;

           input [3:0] d;

           input  load_data, global_reset, clock, increment;

           wire        t1, t2, t3; // internal wires

           wire        see12, reset; // internal wires

 

et_ff etff0(q[0], d[0], increment, load_data, reset, clock);

et_ff etff1(q[1], d[1], t1, load_data, reset, clock);

et_ff etff2(q[2], d[2], t2, load_data, reset, clock);

et_ff etff3(q[3], d[3], t3, load_data, reset, clock);

 

and a1(t1, increment, q[0]); 

and a2(t2, t1, q[1]); 

and a3(t3, t2, q[2]); 

 

         recog12 r1(see12, q);      // счет
          or or1(reset, see12, global_reset); // сброс
 

        endmodule // counter4_bit

 

 module et_ff(q, data, toggle, load_data, reset, clock);

           output q;

           input  data, toggle, load_data, reset, clock;

           wire     m1, m2;

 

           mux mux0(m1, ~q, q, toggle);

           mux mux1(m2, data, m1, load_data);

           dff dff0(q, m2, reset, clock);

 

        endmodule // et_ff

 

        module mux(out, in1, in2, cntrl);

           output out;

           input  in1, in2, cntrl;

 

           assign out = cntrl ? in1 : in2;

 

        endmodule // mux

 

        module dff(q, data, reset, clock);

           output q;

           input  data, reset, clock;

           reg      q;

 

           always @(posedge clock) // at every clock edge, if reset is 1, q is

    if (reset == 1)       


q = 0; 

    else q = data;

 

        endmodule

 

        module recog12(flag, in);

           input [3:0] in;

           output      flag;

 

           assign flag = (in == 4'b1100) ? 1 : 0;

 

        endmodule // recog12

 

        module stumulus;

           wire [3:0] q;

           reg [3:0]  d;

           reg  load_data, global_reset, clk, increment;

 

counter4_bit mod1 (q, d, increment, load_data, global_reset, clk);

 

           initial begin

              global_reset = 0;

              clk = 0;

              increment = 0;

              load_data = 0;

              d = 4'b0100;

 

              #10 global_reset = 1;

              #20 global_reset = 0;

              #20 load_data = 1;

              #20 load_data = 0;

              #20 increment = 1;

              #200 global_reset = 1;

              #20 global_reset = 0;

              #50 load_data = 1;

              #20 load_data = 0;

              #10 increment = 0;

              #20 $finish;

           end // initial begin

 

           always #5 clk = ~clk;

 

      always #10 $display ($time,"  %b %b %b %d -> %b %d",

           increment, load_data, global_reset, d, q, q);

       endmodule // stumulus

 

Начнем подробное рассмотрение этого примера.

 

module counter4_bit(q, d, increment, load_data, global_reset, clock);

output [3:0]     q;

input [3:0] d;

input  load_data, global_reset, clock, increment;

wire        t1, t2, t3; // internal wires

           wire        see12, reset; // internal wires

 

 Как нам уже известно, первые строки модуля содержат его имя и объявление портов. Выход счетчика q, все остальные входные сигналы. t1, t2, t3, see12 и reset объявлены как внутренние цери, подключенные к входам соответствующим субмодулям проекта. 

 

et_ff etff0(q[0], d[0], increment, load_data, reset, clock);

et_ff etff1(q[1], d[1], t1, load_data, reset, clock);

et_ff etff2(q[2], d[2], t2, load_data, reset, clock);

et_ff etff3(q[3], d[3], t3, load_data, reset, clock);

 

В этом фрагменте кода мы определяем 4 счетных триггера с разрешением (Enable/Toggle flip flops)

 

           and a1(t1, increment, q[0]);

           and a2(t2, t1, q[1]);

           and a3(t3, t2, q[2]);
Сигнал увеличения счетчика ,(increment signal) объединяется по И с выходом последнего триггера для переноса сигнала счета на следующий триггер 

 

           recog12 r1(see12, q);

           or or1(reset, see12, global_reset);


Каждый раз когда выход  q изменяется, переменная see12 модифицируется, когда она достигнет 12, внутренний сигнал сброса обеспечит сброс счетчика на очередном такте. Далее рассмотрим поведенческую модель нашего счетчика

 

3.4.30.1. Поведенческая модель (Behavioural Model )

 

Предшествующее описание счетчика было осуществленно на уровне вентилей и триггеров, т.е. на структурном уровне. С другой стороны, при наличии достаточно мощных средств синтеза, поведенческое описание позволяют получить более наглядное описание проекта. Ниже приводится пример поведенческого описания (behavioural description) счетчика 

 

        // 4-bit binary up-counter - of period 13

 

module counter4_bit(q, d, increment, load_data, global_reset, clock);

           output [3:0]     q;

           input [3:0] d;

           input  load_data, global_reset, clock, increment;

 

           reg [3:0]   q;

 

           always @(posedge clock)

              if (global_reset)

              q = 4'b0000;

              else if (load_data)

              q = d;

              else if (increment) begin

              if (q == 12)

                 q = 0;

              else

                 q = q + 1;

              end

 

        endmodule // counter4_bit

 

В отличие от описания на регистровом уровне различие в первых строках только одно – выход q объявлен как регистр

 

Данная модель позволяет путем проведения небольших модификаций получить описание другого типа счетчика  - счетчика в обратном направлении (down counter), Ниже приведен пример такого описания

 

        // 4-bit binary down-counter - of period 13

        module counter4_bit(q, d, decrement, load_data, global_reset, clock);

 

 

        always @(posedge clock)

          if (global_reset)

            q = 12;

          else if (load_data)

            q = d;

          else if (decrement)

            begin

            if (q == 0)

              q = 12;

            else

              q = q - 1;

            end

 

        endmodule
 

Далее рассмотрим реверсивный счетчик (счетчик с переменным направлением счета, Up-Down counter ) с периодом 16.

        // 4-bit binary up-down-counter - of period 16

 module counter4_bit(q, d, updown, count, load_data, global_reset, clock);

 

 

        always @(posedge clock)

          if (reset)

              q = 0;

          else if (load_data)

              q = d;

          else if (count)

              begin

              if (updown)

                  q = q + 1;

              else

                  q = q - 1;

              end

 

        endmodule

 

3.4.31. Временной контроль (Timing Control )

 

В языке Verilog  существуют три варианта осуществления временного контроля (timing control) -  использование задержек (delay), событий (event) и уровни чувствительности (level sensitive)   Рассмотрим их подробнее.

 

3.4.31.1. Задержки (Delay)

Синтакс: 

timing_control_statement::== delay_based statement*

delay_based::== # delay_value 

 

В данном методе вводится задержка между поступлением переменной на вход блока и результатом.

 

        initial begin

          a = 0;      // выполняется в момент t= 0

          #10 b = 2;  // выполняется в момент t= 10

          #15 c = a;  // выполняется в момент t= 25

          #b c = 4;   // выполняется в момент t= 27

          b=5;        // выполняется в момент t= 27

        end

 

Величина задержки может быть определена как постоянная или переменная. Следует помнить, что системное время измеряется в тех единицах, которые установлены в системе. Ниже приводится пример формирования тактового сигнала (creation of a clock signal)

 

        initial begin

          clock = 1'b0;

          forever #5 clock = ~clock;

        end

 

В данном примере каждые 5 единиц времени сигнал переходит из высокого уровня в нзкий и наоборот

 

3.4.31.2. Событийный контроль (Event-Based control)

 

При событийном контроле справедлив следующий синтаксис оператора события (в форме Бэкуса-Науэра)

 

event_control_statement::== 
@ event_identifier 
| @ (event_expression) 
event_expression::== 
| exp 
| event_id 
| posedge exp 
| negedge exp 
| event_exp or event_exp. 

 

При событийном контроле происходит вычисление результата при каждом изменении входных сигналов

 

 

        @ (clock) a = b;// при каждом такте выполняется a = b

 

        @ (negedge clock) a = b;  // когда clock -> 0, то a=b

 

        a = @(posedge clock) b; 
 

3.4.31.3. Защелкивание (Triggers )

Функционирование защелки рассмотрим на примере. 

 

        event data_in; 

        always @(negedge clock)

           if (data[8]==1) -> data_in; 

        always @(data_in) mem[0:1] = buf;

 

В этом примере в каждом отрицательном фронте синхроимпульса, проверяется, равно ли  data[8] единице, если, так то data_in = 1. 

 

3.4.31.4. Список сигналов возбуждения (Sensitivity List )

 

Синтакс: 

event_list_statement::==@ (event_exp or event_exp) 

event_exp::== (event_exp or event_exp). 
 

Если мы желаем, чтобы блок выполнялся  тогда, когда изменилась любая из входных переменных, то список сигналов возбуждения разделяется операторами or 

 

        always @ (i_change or he_changes or she_changes)

             somebody_changed = 1;
 

Здесь изменение хотя бы одной переменной приводит  к изменению остальных

 

3.4.31.5. Задержка распространения в вентиле (Gate Delays )

Синтакс:

delay_statement::== gate_name delay_exp gate_details 

delay_exp::== # delay_time 

 

Данный тип задержки применим только к простейшим вентилям, определяемым в Verilog. 

 

        and #(5) a1(out, in1, in2);

 

3.4.32. Операторы ветвления (Branch Statements )

 

3.4.32.1. Оператор if ( if statement)

 

Синтакс:

 

 if (conditional_expression) statement{ else statement} 

 

Как известно, оператор if (if statement) является оператором услового перехода и выполняется стандартным для языков высокого уровня способом. Условие вычисляется и в зависимости от результата выполняется либо первый оператор, либо второй. Группы операторов  в ветви выделяются операторными скобками (как в языке Паскаль) begin... end. Рассмотрим реализацию нашего мультиплексора с помощью оператора 

If.

 

module multiplexor4_1 (out, in1, in2, in3 ,in4, cntrl1, cntrl2);

       output out;

       input in1, in2, in3, in4, cntrl1, cntrl2;

 

      assign out = (~cntrl2 & ~cntrl1 & in1) | 

                    ( cntrl2 & ~cntrl1 & in2) | 

                    (~cntrl2 &  cntrl1 & in3) |

                    ( cntrl2 &  cntrl1 & in4) ;

endmodule

 

Рассмотрим пример реализация мультиплексора 4 в 1 с помощью логических уравнений. 

 

        module  multiplexor4_1 (out, in1, in2, in3 ,in4, cntrl1, cntrl2);

        output out;

        input in1, in2, in3, in4, cntrl1, cntrl2;

        reg out; 

 

        always @(in1 or in2 or in3 or in4 or cntrl1 or cntrl2)

              if (cntrl1==1)

                 if (cntrl2==1)

                    out = in4;

                 else out = in3;

              else

                 if (cntrl2==1)

                    out = in2;

                 else out = in1;

 

       endmodule

 

 

3.4.32.2. Оператор выбора (case statement)

 

Рассмотренный выше метод реализации мультиплексора удобен при малом числе входов, при их увеличении гораздо разумнее использовать оператор выбора ( case statement), причем его синтаксис аналогичерн оператору выбора в  C. 

 

Синтакс: 

 

conditional::== case (condition) case_item+ endcase 

case_item::== expression+ (seperated by commas) : statement* | default : statement* 

 

Рассмотрим реализацию мультиплексора 4 в 1 с помощью опреатора выбора (case statement).

 

module multiplexor4_1 (out, in1, in2, in3, in4, cntrl1, cntrl2);

          output out;

          input in1, in2, in3, in4, cntrl1, cntrl2;

          reg out; // note must be a register

 

          always @(in1 | in2 | in3 | in4 | cntrl1 | cntrl2)

             case ({cntrl2, cntrl1}) // concatenation

                2'b00 : out = in1;

                2'b01 : out = in2;

                2'b10 : out = in3;

                2'b11 : out = in4;

              default : $display("Please check control bits");

             endcase

        endmodule

 

Вариантами оператора выбора являются операторы casez и casex. 

 

3.4.32.3.Оператор ветвления (conditional operator. ) 

 

Данный оператор по синтаксису анлогичен такому же оператору  в С, так же как и в си этот оператор может быть заменен конструкцией if-else. 

Синтакс: conditional_expression ? true_expression : false_expression 

 

Пример реализации мультиплексора 2 в 1 :

        assign out = control ? in2 : in1;

 

3.4.33. Циклы (Looping Constructs) 

 

В  Verilog существует четыре типа циклов – конструкции  while, for, repeat и forever. Циклы возможно использовать только внутри блоков initial и always. Основное назначение циклов:

while: выполнение операторов пока условие истинноe 

for: в этом цикле возможно инициализировать, проверять и увеличивать индексную переменную явным способом

repeat: цикл с заданным числом повторений

forever: вечный цикл – повторение до конца процесса моделирования

 

3.4.33.1. Цикл While (While Loop )

 

Синтакс:

 looping_statement::== while (conditional) statement 

 

Цикл  while выполняется до тех пор, пока условие истинно, в качестве условия может быть использовано любое логическое выражение. Операторы в цикле группируются с использованием операторных скобок  begin ... end. Иже приведен пример while. 

 

        /* How many cups of volume 33 ml does it

           take to fill a jug of volume 1 litre  */

 

        `define JUGVOL 1000

        `define CUPVOL 33

 

        module cup_and_jugs;

           integer cup, jug;

 

           initial begin

              cup = 0; jug = 0;

              while (jug < `JUGVOL) begin

              jug = jug + `CUPVOL;

              cup = cup + 1;

              end // while (jug < JUG_VOL)

 

              $display("It takes %d cups", cup);

           end // initial begin

 

        endmodule // cup_and_jugs

Notice the use of the "define" statement. 

 

3.4.33.2. Цикл For  (For Loop )

 

Синтакс 

 

for (reg_initialisation ; conditional ; reg_update) statement 

 

Цикл for в Verilog  аналогичен циклу for в языке C.  Ниже приведен пример его использования.

 

 

        integer list [31:0];

        integer index;

 

        initial begin

           for(index = 0; index < 32; index = index + 1)

              list[index] = index + index;

           end

 

3.4.33.3. Цикл Repeat (Repeat Loop )

 

Синтакс:

 looping_statement::== repeat (conditional) statement 

 

Цикл  repeat выполняется конечное число раз, условие (conditional) можеит быть константой, переменной или сигналом, но обязательно быть целым числом.  Ниже приведен пример цикла repeat
 

        module comply;

          int count;

        initial begin

          count = 128;

          repeat (count) begin

             $display("%d seconds to comply", count);

             count = count - 1;

          end

        end

 

        endmodule


Note: The loop will execute 128 times regardless of whether the value of count changes after entry to the loop

 

3.4.33.4. Вечный цикл (Forever Loop )

 

Синтаксис:

 forever statement 

 

Цикл  forever выполняется непрерывно, пока моделирование не будет остановлено командой $finish. Ниже приведен пример вечного цикла

           reg clock;

 

           initial begin

              clock = 1'b0;


forever #5 clock = ~clock; 



end

 

           initial #2000 $finish;
 

3.4.34. Работа с файлами в Verilog
3.4.34.1. Открытие файла (Opening a file )

 

Ключевые слова: $open 

Синтакс: <file_handle> = $fopen("<file_name"); 

 

В процессе моделирования в языке Verilog имеется возможность записывать результаты прогона в файл, что в дальнейшем позволяет провести его более подробный анализ

Ниже приведен пример открытия файлов нва запись 

 

        integer handleA, handleB;

 // an integer has 32 bits so is perfect

// for this usage

 

        initial begin

          handleA = $fopen("myfile.out);

// handleA = 0000_0000_0000_0000_0000_0000_0000_0010

 handleB = $fopen("anotherfile.out");

// handleB = 0000_0000_0000_0000_0000_0000_0000_0100

        end

 

Когда файл открыт, его можно использовать для записи. Всего допустимо открыть не более 31 файла одновременно. 

 

3.4.34.2. Запись в файл (Writing to a file )

 

Ключевые слова: $fdisplay, $fmonitor, $fwrite 
Синтакс:

$fdisplay(<file_handles>, --same as display--); 
$fmonitor(<file_handles>, --same as monitor--); 
$fwrite(<file_handles>, --same as write--); 

 

Для  записиможет быть открыто несколько файлов,  ниже приведен пример записи числа  0000_0000_0000_0000_0000_0000_0000_0011 в файлы.

 

        integer channelsA;

 

        initial begin

          channelsA = handleA | 1;

$fdisplay(channelsA, "Hello"); 

        end

 

3.4.34.3.Закрытие файла (Closing a file )

 

Ключевое слово: $fclose

Синтакс: $fclose(handle); 

 

Когда используется несколько файлов, то обычно файлы, которые уже не требуются закрывают для облегчения жизни несчастной операционной системе. Для этого и служит директива $fclose, после закрытия работа с данным  файлом естественно невозможна. 

 

3.4.35. Инициализация регистровых файлов (памяти) (Initialising Memories )

Ключевые слова: $readmemb, $readmemh 

Синтакс: 

$readmemb("<file_name>", <memory_name>"); 

$readmemb("<file_name>", <memory_name>, memory_start"); 

$readmemb("<file_name>", <memory_name>, memory_start, memory_finish"); 

 

Для инициализации регистровых файлов  используется следующая конструкция. Содержимое регистрового файла храниться в файле с форматом, приведенном ниже, все адреса 16-ричные

 

        @003

        00000011

        00000100

        00000101

        00000110

        00000111

        00001000

        00001001


Для улучшения читаемости возможна и такая запись

        @003

        00000011

        00000100

        00000101

        @006

        00000110

        00000111

        @008

        00001000

        00001001

 

Пример инициализации памяти приведен ниже 

 

 

        module testmemory;

           reg [7:0] memory [9:0];

           integer   index;

 

           initial begin

              $readmemb("mem.dat", memory);

 

              for(index = 0; index < 10; index = index + 1)

              $display("memory[%d] = %b", index[4:0], memory[index]);

           end

        endmodule // testmemory


Содержимое файла файла mem.data имеет вид 

 
        1000_0001

        1000_0010

        0000_0000

        0000_0001

        0000_0010

        0000_0011

        0000_0100

        0000_0101

        0000_0110

        0000_0000

 

3.4.36. Задание векторов входных сигналов для моделирования (Verilog Input Vectors )

Для задания входных векторов для моделирования системы требуется весьма ограниченное подмножество языка Verilog. Рассмотрим простой пример

        initial

        begin

           clk = 1'b0;

           globalReset = 1'b1;

           in = 1'b1;

        end
 

Выражение 1'b0 предсавляет собой двоичное представление 0. 

Существуют две основных части : 

 

        initial

        begin

             -----

        end

 

        always

        begin

             -----

        end

Блок initial начинает выполняться с началом моделирования (simulation), поэтому в нем инициализируются сигналы и определяется последовательность их изменений

Тактовый сигнал существует в течении всего времени моделирования. Задержка между моментами смены сигналов (событиями, events) определяется с помощюь знака  # . Например
 

           #160 globalReset = 0;

           #160 in = 0;

           #160 in = 1;

           #320 in = 0;


В приведенном примере после 160 единиц времени ожидания сигнал globalReset устанавливается в 0, затем 160 единиц ожидания и установка сигнала in в 0 и т.д.

Наиболее часто конструкция always используется для задания тактового сигнала. Например: 

 

        always

        begin

           #10 clk = ~clk;

        end


Здесь сигнал clk должен быть проинвертирован каждые 10 единиц измерения времени, что по сути означает, что период тактового сигнала - 20 единиц измерения времени (time units). 

Кроме того, в процессе моделирования неплохо использовать цикл repeat для задания заданного числа повторений

 

        repeat(10)

        begin

            #40 in = 0;

            #20 in = 1;

        end
 

Для генерации случайных чисел в диапазоне  [0, n-1] используется следующая команда

 

        {$random} % n

Комбинируя эти команды создадим генератор пачек из 14 случайных чисел каждые 20 единиц времени (time units)

 

        repeat(14)

        begin

            #20 in = {$random} / 16  % 2;

        end

 

